Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

Profiling

In[∘]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[∘]= Show Profile Monitor

External Utilities

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CCF[ℰ_] := PP_CCF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
      Expand[ℰ] //. ⅇ^x_ ⅇ^y_ :> ⅇ^(x+y) /. ⅇ^x_ :> ⅇ^CCF[x]]];
CF[ℰ_List] := CF /@ ℰ;
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@Module[
    {vs = Cases[ℰ, (y | b | t | a | x | η | β | τ | α | ξ)_, ∞] ⋃ {y, b, t, a, x, η, β, τ, α, ξ}},
    Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :> CCF[c] (Times @@ vs^ps)]
   ];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; 𝔼[*L*, *Q*, *P*] stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_{$k_} := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Program

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z};  (u_{i_})* := (u*)_i;
```

Program

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PP_Collect@Collect[ℰ, ζ];
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := PP_Zip[
  (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. ⧴ ∂_{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the “*Q*-level zips” on 𝔼(*L*, *Q*, *P*) = Pe^{L+Q}. Such zips regard the *L* variables as scalars.

Program

In[∘]:=
```
QZip_{ζs_List}@𝔼[L_, Q_, P_] := PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    Q2 = CF[(Q1 = CF[c + ηs.zs /. zrule]) /. Alternatives @@ zs → 0];
    CF /@ 𝔼[L, Q2, Det[qt] e^{-Q2} Zip_{ζs}[e^{Q1} (P /. zrule)]] ];
```

In[∘]:=
```
$T = 1;
QZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  PP_{QZip}@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2, tin = TimeUsed[], out},
    zs = Table[ζ*, {ζ, ζs}];
    c = CF[Q /. Alternatives @@ (ζs ⋃ zs) → 0];
    ys = CF@Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = CF@Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = CF@Inverse@Table[K δ_{z,ζ*} - ∂_{z,ζ} Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → CF[qt.(zs + ys)]];
    Q2 = CF[(Q1 = CF[c + ηs.zs /. zrule]) /. Alternatives @@ zs → 0];
    out = CF /@ 𝔼[L, Q2, Det[qt] e^{-Q2} Zip_{ζs}[e^{Q1} (P /. zrule)]];
    (*If[(tin=TimeUsed[]-tin)>$T, $T=2tin; Echo[{tin,ζs,L,Q,P}]];*) out ];
```

Program

Upper to lower and lower to Upper:

Program

In[∘]:=
```
U2l = {B_{i_}^{p_.} → e^{-p ħ γ b_i}, B^{p_.} → e^{-p ħ γ b}, T_{i_}^{p_.} → e^{p ħ t_i}, T^{p_.} → e^{p ħ t}, 𝒜_{i_}^{p_.} → e^{p γ α_i}, 𝒜^{p_.} → e^{p γ α}};
l2U = {e^{c_. b_{i_}+d_.} :→ B_i^{-c/(ħ γ)} e^d, e^{c_. b+d_.} :→ B^{-c/(ħ γ)} e^d,
    e^{c_. t_{i_}+d_.} :→ T_i^{c/ħ} e^d, e^{c_. t+d_.} :→ T^{c/ħ} e^d,
    e^{c_. α_{i_}+d_.} :→ 𝒜_i^{c/γ} e^d, e^{c_. α+d_.} :→ 𝒜^{c/γ} e^d,
    e^{ℰ_} :→ e^{Expand@ℰ}};
```

Program

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single"P". Here the z's are b and α and the ζ's are β and a.

In[∘]:=
```
LZipMaxT = 0;
LZip_{ζs_List}@𝔼[L_, Q_, P_] :=
  PP_{LZip}@Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2, start = TimeUsed[], out},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[K δ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    out = CF /@ (CF /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_{ζs}[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U);
    If[(tin = TimeUsed[] - start) > LZipMaxT, LZipMaxT = tin; LZipArgs = {tin, ζs, L, Q, P}];
    out
   ];
```

Program

```
In[]:= (*LZip_{ζs_List}@𝔼[L_,Q_,P_] := PP_{LZip}@Module[{ζ,z,zs,c,ys,ηs,lt,zrule,L1,L2,Q1,Q2},
    zs=Table[ζ*,{ζ,ζs}];
    c=L/.Alternatives@@(ζs⋃zs)→0;
    ys=Table[∂_ζ(L/.Alternatives@@zs→0),{ζ,ζs}];
    ηs=Table[∂_z(L/.Alternatives@@ζs→0),{z,zs}];
    lt=Inverse@Table[Kδ_{z,ζ*}-∂_{z,ζ}L,{ζ,ζs},{z,zs}];
    zrule=Thread[zs→lt.(zs+ys)];
    L2=(L1=c+ηs.zs/.zrule)/.Alternatives@@zs→0;
    Q2=(Q1=Q/.U2l/.zrule)/.Alternatives@@zs→0;
    CF /@ (CF/@𝔼[L2,Q2,Det[lt]ⅇ^{-L2-Q2}Zip_{ζs}[ⅇ^{L1+Q1}(P/.U2l/.zrule)]]//.l2U) ];*)
```

Program

```
In[]:= B_{}[L_, R_] := L R;
B_{is__}[L_𝔼, R_𝔼] := PP_B@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Join@@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Join@@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{is___}[L_, R_] := B_{{is}}[L, R];
```

Program

𝔼 morphisms with domain and range.

Program

```
In[]:= B_{is_List}[𝔼_{d1_→r1_}[L1_, Q1_, P1_], 𝔼_{d2_→r2_}[L2_, Q2_, P2_]] :=
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ B_{is}[𝔼[L1, Q1, P1], 𝔼[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] // 𝔼_{d2_→r2_}[L2_, Q2_, P2_] :=
  B_{r1⋂d2}[𝔼_{d1→r1}[L1, Q1, P1], 𝔼_{d2→r2}[L2, Q2, P2]];
𝔼_{d1_→r1_}[L1_, Q1_, P1_] ≡ 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[L1, Q1, P1] ≡ 𝔼[L2, Q2, P2]);
𝔼_{d1_→r1_}[L1_, Q1_, P1_] 𝔼_{d2_→r2_}[L2_, Q2_, P2_] ^:=
  𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[L1, Q1, P1] 𝔼[L2, Q2, P2]);
𝔼_{d_→r_}[L_, Q_, P_]_{$k_} := 𝔼_{d→r} @@ 𝔼[L, Q, P]_{$k};
𝔼_[ℰ___][i_] := {ℰ}〚i〛;
```

Program

“Define” code

Program

Define[lhs = rhs,   ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

In[∘]:=
```
SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_is__ = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
   ReleaseHold[Hold[
      SD[op_{nisp,$k_Integer}, PP_Boot@Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
      SD[op_isp, op_{{is},$k}]; SD[op_sis__, op_{sis}];
     ] /. {SD → SetDelayed,
      isp → {is} /. {i → i_, j → j_, k → k_},
      nis → {is} /. {i → ii, j → jj, k → kk},
      nisp → {is} /. {i → ii_, j → jj_, k → kk_}
     }] ]]
```

Program

Booting Up

Program

In[∘]:=
```
$k = 2; (*ℏ=γ=1;*)
```

Program

In[∘]:=
```
Define[am_{i,j→k} = 𝔼_{{i,j}→{k}}[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
 bm_{i,j→k} = 𝔼_{{i,j}→{k}}[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i}-1) η_j y_k}]_{$k}]
```

Program

In[∘]:=
```
Define[R_{i,j} = 𝔼_{{}→{i,j}}[ℏ a_j b_i, ℏ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ ϵ ℏ})^k (ℏ y_i x_j)^k)/(k (1 - e^{k γ ϵ ℏ})))]_{$k},
 R̄_{i,j} = 𝔼_{{}→{i,j}}[-ℏ a_j b_i, -ℏ x_j y_i / B_i, 1 + If[$k == 0, 0, (R̄_{{i,j},$k-1})_{$k}[3] -
      (((R̄_{{i,j},0})_{$k} R_{1,2} (R̄_{{3,4},$k-1})_{$k}) // (bm_{i,1→i} am_{j,2→j}) // (bm_{i,3→i} am_{j,4→j}))[3]]],
 P_{i,j} = 𝔼_{{i,j}→{}}[β_i α_j / ℏ, η_i ξ_j / ℏ, 1 + If[$k == 0, 0, (P_{{i,j},$k-1})_{$k}[3] -
      (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))[3]]]]
```

Program

In[∘]:=
```
Define[aS_j = R̄_{i,j} ~ B_i ~ P_{i,j},
 aS̄_i = 𝔼_{{i}→{i}}[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0, (aS̄_{{i},$k-1})_{$k}[3] -
      ((aS̄_{{i},0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{{i},$k-1})_{$k})[3]]]]
```

Program

In[∘]:=
```
Define[bS_i = R_{i,1} ~ B_1 ~ aS_1 ~ B_1 ~ P_{i,1},
 bS̄_i = R_{i,1} ~ B_1 ~ aS̄_1 ~ B_1 ~ P_{i,1},
 aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
 bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

In[∘]:=

```
Define[dm_{i,j→k} = (𝔼_{{i,j}→{i,j}}[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1]
    (aΔ_{i→1,2} // aΔ_{2→2,3} // a̅S̅_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) // (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼_{{i}→{1,2}}[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (b̅S̅_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

In[∘]:=

```
Define[C_i = 𝔼_{{}→{i}}[0, 0, B_i^{1/2} ⅇ^{-ħ ϵ a_i/2}]_{$k},
  C̅_i = 𝔼_{{}→{i}}[0, 0, B_i^{-1/2} ⅇ^{ħ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} C̅_2) // dm_{1,2→1} // dm_{1,3→i},
  K̅i̅n̅k̅_i = (R̅_{1,3} C_2) // dm_{1,2→1} // dm_{1,3→i}]
```

Program

Note. $t == \epsilon \mathrm{a} - \gamma \mathrm{b}$ and $b == -t/\gamma + \epsilon \mathrm{a}/\gamma$.

Program

In[∘]:=

```
Define[b2t_i = 𝔼_{{i}→{i}}[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, ⅇ^{ϵ β_i a_i/γ}]_{$k},
  t2b_i = 𝔼_{{i}→{i}}[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, ⅇ^{ϵ τ_i a_i}]_{$k}]
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$am \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}_{\{i,j\}\to\{k\}}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}_{\{i,j\}\to\{k\}}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ y_k\,\eta_i+\frac{y_k\,\eta_j}{\mathcal{A}_i}+\frac{x_k\,\xi_i}{\mathcal{A}_j}+\frac{(1-B_k)\,\eta_j\,\xi_i}{\hbar}+x_k\,\xi_j,$$
$$1+\left(-\frac{y_k\,\beta_i\,\eta_j}{\mathcal{A}_i}-\frac{x_k\,\beta_j\,\xi_i}{\mathcal{A}_j}+a_k\,B_k\,\eta_j\,\xi_i+\frac{\gamma\,\hbar\,x_k\,y_k\,\eta_j\,\xi_i}{\mathcal{A}_i\,\mathcal{A}_j}+\frac{(\gamma-3\,\gamma\,B_k)\,y_k\,\eta_j^2\,\xi_i}{2\,\mathcal{A}_i}+\frac{(\gamma-3\,\gamma\,B_k)\,x_k\,\eta_j\,\xi_i^2}{2\,\mathcal{A}_j}+\frac{\left(\gamma-4\,\gamma\,B_k+3\,\gamma\,B_k^2\right)\,\eta_j^2\,\xi_i^2}{4\,\hbar}\right)\epsilon+$$
$$O[\epsilon]^2\Big]$$

$$R \to \mathbb{E}_{\{\}\to\{i,j\}}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}_{\{\}\to\{i,j\}}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1+\left(-\frac{\hbar^2\,a_j\,x_j\,y_i}{B_i}-\frac{3\,\gamma\,\hbar^3\,x_j^2\,y_i^2}{4\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}_{\{i,j\}\to\{\}}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar}+O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\left(\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}_{\{i\}\to\{i\}}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\left(\frac{\gamma\,\hbar\,y_i\,\eta_i}{B_i}-\frac{y_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\eta_i^2}{2\,B_i^2}\right)\epsilon+O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}_{\{i\}\to\{i\}}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ -\frac{y_i\,\mathcal{A}_i\,\eta_i}{B_i}-x_i\,\mathcal{A}_i\,\xi_i+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1+\Big(\frac{\gamma\,\hbar\,y_i\,\mathcal{A}_i\,\eta_i}{B_i}-\frac{y_i\,\mathcal{A}_i\,\beta_i\,\eta_i}{B_i}-\frac{\gamma\,\hbar\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2}{2\,B_i^2}-\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-x_i\,\mathcal{A}_i\,\beta_i\,\xi_i+\frac{a_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{B_i}-$$
$$\frac{\gamma\,\hbar\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i}{B_i}+\frac{(-\gamma\,\mathcal{A}_i+\gamma\,B_i\,\mathcal{A}_i)\,\eta_i\,\xi_i}{B_i}+\frac{(\mathcal{A}_i-B_i\,\mathcal{A}_i)\,\beta_i\,\eta_i\,\xi_i}{\hbar\,B_i}+\frac{y_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i}{2\,B_i^2}-$$
$$\frac{1}{2}\,\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+\frac{x_i\left(3\,\gamma\,\mathcal{A}_i^2-\gamma\,B_i\,\mathcal{A}_i^2\right)\eta_i\,\xi_i^2}{2\,B_i}+\frac{\left(-3\,\gamma\,\mathcal{A}_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\right)\eta_i^2\,\xi_i^2}{4\,\hbar\,B_i^2}\Big)\epsilon+O[\epsilon]^2\Big]$$

$$a\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\left(-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$b\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$d\Delta \to \mathbb{E}_{\{i\}\to\{j,k\}}\Big[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\ y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,$$
$$1+\left(\frac{1}{2}\,\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2-\hbar\,a_j\,x_k\,\xi_i+\frac{1}{2}\,\gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

$$C \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{C} \to \mathbb{E}_{\{\}\to\{i\}}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}_{\{\}\to\{i\}}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\left(\frac{\hbar\,a_i}{2\,\sqrt{B_i}}-\frac{\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,\sqrt{B_i}}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}_{\{\}\to\{i\}}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i}+\left(-\frac{1}{2}\,\hbar\,a_i\,\sqrt{B_i}-\frac{\hbar^2\,a_i\,x_i\,y_i}{\sqrt{B_i}}-\frac{3\,\gamma\,\hbar^3\,x_i^2\,y_i^2}{4\,B_i^{3/2}}\right)\epsilon+O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}_{\{i\}\to\{i\}}\left[a_i\,\alpha_i-\gamma\,b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@
 {{"[a,x]" → ((𝔼{}→{1,2}[0, 0, a2 x1] // am1,2→1)[3] - (𝔼{}→{1,2}[0, 0, a1 x2] // am1,2→1)[3]),
    "[b,y]" → ((𝔼{}→{1,2}[0, 0, y2 b1] // bm1,2→1)[3] - (𝔼{}→{1,2}[0, 0, y1 b2] // bm1,2→1)[3])} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼{}→{1}[0, 0, y1] ~B1~ bΔ1→1,2],
   "Δ[b]" → Last[𝔼{}→{1}[0, 0, b1] ~B1~ bΔ1→1,2],
   "Δ[a]" → Last[𝔼{}→{1}[0, 0, a1] ~B1~ aΔ1→1,2],
   "Δ[x]" → Last[𝔼{}→{1}[0, 0, x1] ~B1~ aΔ1→1,2]},
  {
    "S(a)" → ((𝔼{}→{1}[0, 0, a1] ~B1~ aS1)[3]),
    "S(x)" → ((𝔼{}→{1}[0, 0, x1] ~B1~ aS1)[3]),
    "S(b)" → ((𝔼{}→{1}[0, 0, b1] ~B1~ bS1)[3]),
    "S(y)" → ((𝔼{}→{1}[0, 0, y1] ~B1~ bS1)[3])
   } /. z_1 → z}
```

Out[∘]= $\{1.39063, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\}, \{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1+b_2) + O[\epsilon]^3,\ \Delta[a] \to (a_1+a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1+x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\}, \{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am1,2→1 // am1,3→1) ≡ (am2,3→2 // am1,2→1), (bm1,2→1 // bm1,3→1) ≡ (bm2,3→2 // bm1,2→1)}
 ]
```

Out[∘]= {0.234375, {True, True}}

R and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {Ri,j, Pi,k, HL[(Ri,j // Pi,k) ≡ 𝔼{k}→{j}[aj αk, xj ξk, 1]]}]
```

Out[∘]= $\{0.15625, \{\mathbb{E}_{\{\}\to\{i,j\}}[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma\hbar^3 x_j^2 y_i^2)\epsilon + (\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4)\epsilon^2 + \frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4],$
$\mathbb{E}_{\{i,k\}\to\{\}}[\frac{\alpha_k\beta_i}{\hbar}, \frac{\eta_i\xi_k}{\hbar}, 1 + \frac{\gamma\eta_i^2\xi_k^2\epsilon}{4\hbar} + \frac{1}{288\hbar^2}(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2 + (\frac{1}{24}\gamma^3\hbar\eta_i^2\xi_k^2 + \frac{1}{6}\gamma^3\eta_i^3\xi_k^3 + \frac{13\gamma^3\eta_i^4\xi_k^4}{96\hbar} + \frac{5\gamma^3\eta_i^5\xi_k^5}{144\hbar^2} + \frac{\gamma^3\eta_i^6\xi_k^6}{384\hbar^3})\epsilon^3 + O[\epsilon]^4], \text{True}\}\}$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

In[∘]:=
```
Timing[HL /@ {(aS̄1 // aS1) ≡ 𝔼{1}→{1}[a1 α1, x1 ξ1, 1], (bS̄1 // bS1) ≡ 𝔼{1}→{1}[b1 β1, y1 η1, 1]}]
```

Out[∘]= {0.53125, {True, True}}

(co)-associativity on both sides

In[∘]:=
```
Timing[
 HL /@ {(aΔ_{1→1,2} // aΔ_{2→2,3}) ≡ (aΔ_{1→1,3} // aΔ_{1→1,2}), (bΔ_{1→1,2} // bΔ_{2→2,3}) ≡ (bΔ_{1→1,3} // bΔ_{1→1,2}),
   (am_{1,2→1} // am_{1,3→1}) ≡ (am_{2,3→2} // am_{1,2→1}), (bm_{1,2→1} // bm_{1,3→1}) ≡ (bm_{2,3→2} // bm_{1,2→1})}]
```

Out[∘]= {0.578125, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing[HL /@ {(am_{1,2→1} // aΔ_{1→1,2}) ≡ ((aΔ_{1→1,3} aΔ_{2→2,4}) // (am_{3,4→2} am_{1,2→1})),
   (bm_{1,2→1} // bΔ_{1→1,2}) ≡ ((bΔ_{1→1,3} bΔ_{2→2,4}) // (bm_{3,4→2} bm_{1,2→1}))}]
```

Out[∘]= {0.78125, {True, True}}

An explicit formula for aS_i

In[∘]:=
```
Timing@Block[{$k = 4}, HL[aS_i ≡ (𝔼_{{i}→{i,j}}[-α_i a_j, -ξ_i x_i,
      Sum[Expand[(e^{ξ_i x_i} (-ħ γ ϵ)^k)/(2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, e^{-ξ_i e^{ħ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k} //
    am_{i,j→i})]]
```

Out[∘]= {4.45313, True}

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{1,2→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{1,2→1},
   (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{1,2→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{1,2→1}}]
```

Out[∘]= {0.921875, {True, True, True, True}}

But not with the opposite product:

In[∘]:=
```
Timing[Short[# ≡ 𝔼_{{1}→{1}}[0, 0, 1]] & /@ {
   (aΔ_{1→1,2} ~ B_1 ~ aS_1) ~ B_{1,2} ~ am_{2,1→1}, (aΔ_{1→1,2} ~ B_2 ~ aS_2) ~ B_{1,2} ~ am_{2,1→1},
   (bΔ_{1→1,2} ~ B_1 ~ bS_1) ~ B_{1,2} ~ bm_{2,1→1}, (bΔ_{1→1,2} ~ B_2 ~ bS_2) ~ B_{1,2} ~ bm_{2,1→1}}]
```

Out[∘]= $\Big\{1.35938, \Big\{\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1 \mathcal{A}_1 \xi_1+\gamma^2\epsilon^2\hbar^2 x_1\mathcal{A}_1\xi_1-2\gamma\epsilon^2\hbar^2 a_1 x_1\mathcal{A}_1\xi_1+2\gamma^2\epsilon^2\hbar^2 x_1^2\mathcal{A}_1^2\xi_1^2\right)=0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar x_1\xi_1-\gamma^2\epsilon^2\hbar^2 x_1\xi_1+2\gamma^2\epsilon^2\hbar^2 x_1^2\xi_1^2\right)=0,$

$\frac{1}{2}\left(-2\gamma\epsilon\hbar y_1\eta_1-\gamma^2\epsilon^2\hbar^2 y_1\eta_1+2\gamma^2\epsilon^2\hbar^2 y_1^2\eta_1^2\right)=0,$

$\frac{1}{2B_1^2}\left(-2\gamma\epsilon\hbar B_1 y_1\eta_1+\gamma^2 \ll 4 \gg \eta_1-\ll 1 \gg+2\gamma^2\epsilon^2\hbar^2 y_1^2\eta_1^2\right)=0\Big\}\Big\}$

S is an algebra anti-(co)morphism

In[∘]:=
```
Timing[HL /@ {am_{1,2→1} ~ B_1 ~ aS_1 ≡ (aS_1 aS_2) ~ B_{1,2} ~ am_{2,1→1}, bm_{1,2→1} ~ B_1 ~ bS_1 ≡ (bS_1 bS_2) ~ B_{1,2} ~ bm_{2,1→1},
   aS_1 ~ B_1 ~ aΔ_{1→1,2} ≡ aΔ_{1→2,1} ~ B_{1,2} ~ (aS_1 aS_2), bS_1 ~ B_1 ~ bΔ_{1→1,2} ≡ bΔ_{1→2,1} ~ B_{1,2} ~ (bS_1 bS_2)}]
```

Out[∘]= {1.42188, {True, True, True, True}}

Pairing axioms

In[∘]:=
```
Timing[HL /@ {(bm_{1,2→1} 𝔼_{{3}→{3}}[α_3 a_3, ξ_3 x_3, 1]) ~ B_{1,3} ~ P_{1,3} ≡
    (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] 𝔼_{{2}→{2}}[β_2 b_2, η_2 y_2, 1] aΔ_{3→4,5}) ~ B_{1,4} ~ P_{1,4} ~ B_{2,5} ~ P_{2,5},
  (bΔ_{1→1,2} 𝔼_{{3}→{3}}[α_3 a_3, ξ_3 x_3, 1] 𝔼_{{4}→{4}}[α_4 a_4, ξ_4 x_4, 1]) ~ B_{1,3} ~ P_{1,3} ~ B_{2,4} ~ P_{2,4} ≡
    (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] am_{3,4→3}) ~ B_{1,3} ~ P_{1,3} }]
```
Out[∘]= {0.640625, {True, True}}

In[∘]:=
```
Timing[HL /@ {((bS_1 𝔼_{{2}→{2}}[α_2 a_2, ξ_2 x_2, 1]) // P_{1,2}) ≡ ((𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] aS_2) // P_{1,2}),
  (overline{bS}_1 𝔼_{{2}→{2}}[α_2 a_2, ξ_2 x_2, 1]) ~ B_{1,2} ~ P_{1,2} ≡ (𝔼_{{1}→{1}}[β_1 b_1, η_1 y_1, 1] overline{aS}_2) ~ B_{1,2} ~ P_{1,2}}]
```
Out[∘]= {0.421875, {True, True}}

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
Timing@{{
      "[a,y]" →
       ((𝔼_{{}→{1,2}}[0, 0, y_2 a_1] ~ B_{1,2} ~ dm_{1,2→1}) [3] - (𝔼_{{}→{1,2}}[0, 0, y_1 a_2] ~ B_{1,2} ~ dm_{1,2→1}) [3]),
      "[b,x]"  → ((𝔼_{{}→{1,2}}[0, 0, x_2 b_1] ~ B_{1,2} ~ dm_{1,2→1}) [3] -
         (𝔼_{{}→{1,2}}[0, 0, x_1 b_2] ~ B_{1,2} ~ dm_{1,2→1}) [3]),
      "xy-qyx" → ((𝔼_{{}→{1,2}}[0, 0, x_1 y_2] ~ B_{1,2} ~ dm_{1,2→1}) [3] -
         (1 + ϵ) (𝔼_{{}→{1,2}}[0, 0, y_1 x_2] ~ B_{1,2} ~ dm_{1,2→1}) [3])
     } /. {z_1 → z} // Expand // Factor,
   {
    "Δ(a)" → ((𝔼_{{}→{1}}[0, 0, a_1] ~ B_1 ~ dΔ_{1→1,2}) [3]),
    "Δ(x)" → ((𝔼_{{}→{1}}[0, 0, x_1] ~ B_1 ~ dΔ_{1→1,2}) [3]),
    "Δ(b)" → ((𝔼_{{}→{1}}[0, 0, b_1] ~ B_1 ~ dΔ_{1→1,2}) [3]),
    "Δ(y)" → ((𝔼_{{}→{1}}[0, 0, y_1] ~ B_1 ~ dΔ_{1→1,2}) [3])
   } // Simplify,
   {
     "S(a)" → ((𝔼_{{}→{1}}[0, 0, a_1] ~ B_1 ~ dS_1) [3]),
     "S(x)" → ((𝔼_{{}→{1}}[0, 0, x_1] ~ B_1 ~ dS_1) [3]),
     "S(b)" → ((𝔼_{{}→{1}}[0, 0, b_1] ~ B_1 ~ dS_1) [3]),
     "S(y)" → ((𝔼_{{}→{1}}[0, 0, y_1] ~ B_1 ~ dS_1) [3])
    } /. {z_1 → z} // Simplify
  }
```
Out[∘]=

$$\Big\{10.125,\ \Big\{\Big\{[a,y] \to -y\,\gamma + O[\epsilon]^3,\ [b,x] \to x\,\epsilon + O[\epsilon]^3,$$

$$xy\text{-}qyx \to \frac{1-B}{\hbar} + (a\,B - x\,y + x\,y\,\gamma\,\hbar)\,\epsilon + \left(-\frac{1}{2}a^2\,B\,\hbar + \frac{1}{2}x\,y\,\gamma^2\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3\Big\},$$

$$\Big\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar\,a_1\,x_2\,\epsilon + \frac{1}{2}\hbar^2\,a_1^2\,x_2\,\epsilon^2 + O[\epsilon]^3,$$

$$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^3\Big\},$$

$$\Big\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a\,x\,\hbar\,\epsilon - \frac{1}{2}\left(a^2\,x\,\hbar^2\right)\epsilon^2 + O[\epsilon]^3,$$

$$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\,\gamma\,\hbar\,\epsilon}{B} - \frac{\left(y\,\gamma^2\,\hbar^2\right)\epsilon^2}{2\,B} + O[\epsilon]^3\Big\}\Big\}\Big\}$$

(co)-associativity

```
In[∘]:= Timing[
  HL /@ {(dΔ1→1,2 // dΔ2→2,3) ≡ (dΔ1→1,3 // dΔ1→1,2), (dm1,2→1 // dm1,3→1) ≡ (dm2,3→2 // dm1,2->1)}]
Out[∘]= {6.25, {True, True}}
```

Δ is an algebra morphism

```
In[∘]:= Timing@HL[dm1,2→1 ~ B1 ~ dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~ B1,2,3,4 ~ (dm3,4→2 dm1,2→1)]
Out[∘]= {7.4375, True}
```

S_2 inverts R, but not S_1:

```
In[∘]:= Timing@{R1,2 ~ B1 ~ dS1 ≡ R̄1,2, HL[R1,2 ~ B2 ~ dS2 ≡ R̄1,2]}
```

Out[∘]= $\left\{0.984375, \left\{\frac{1}{4 B_1^3}\left(4 \gamma \epsilon \hbar^2 B_1^2 x_2 y_1 - 2 \gamma^2 \epsilon^2 \hbar^3 B_1^2 x_2 y_1 + 4 \gamma \epsilon^2 \hbar^3 a_2 B_1^2 x_2 y_1 + 8 \gamma^2 \epsilon^2 \hbar^4 B_1 x_2^2 y_1^2 - 4 \gamma \epsilon^2 \hbar^4 a_2 B_1 x_2^2 y_1^2 - 3 \gamma^2 \epsilon^2 \hbar^5 x_2^3 y_1^3\right) = 0, \text{True}\right\}\right\}$

S is convolution inverse of id

```
In[∘]:= Timing[HL[# ≡ 𝔼{1}→{1}[0, 0, 1]] & /@
   {(dΔ1→1,2 ~ B1 ~ dS1) ~ B1,2 ~ dm1,2→1, (dΔ1→1,2 ~ B2 ~ dS2) // dm1,2→1}]
Out[∘]= {11.5, {True, True}}
```

S is a (co)-algebra anti-morphism

```
In[∘]:= Timing[HL /@
   Expand /@ {dm1,2→1 ~ B1 ~ dS1 ≡ (dS1 dS2) ~ B1,2 ~ dm2,1→1, dS1 ~ B1 ~ dΔ1→1,2 ≡ dΔ1→2,1 ~ B1,2 ~ (dS1 dS2)}]
Out[∘]= {20.7344, {True, True}}
```

Quasi-triangular axiom 1:

```
In[∘]:= Timing@HL[R1,2 ~ B1 ~ dΔ1->1,3 ≡ (R1,4 R3,2) ~ B2,4 ~ dm2,4→2]
Out[∘]= {0.796875, True}
```

Quasi-triangular axiom 2:

```
In[∘]:= Timing@HL[((dΔ1->1,2 R3,4) ~ B1,2,3,4 ~ (dm1,3→1 dm2,4→2)) ≡ ((dΔ1→2,1 R3,4) ~ B1,2,3,4 ~ (dm3,1→1 dm4,2→2))]
Out[∘]= {7.85938, True}
```

The Drinfel'd element inverse property, $(u_1\,\bar{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to 1} \equiv \mathbb{E}[0, 0, 1]$:

```
In[∘]:= Timing@HL[((R1,2 ~ B1 ~ dS1 ~ B1,2 ~ dm2,1→i) (R1,2 ~ B2 ~ dS2 ~ B2 ~ dS2 ~ B1,2 ~ dm2,1→j)) ~ Bi,j ~ dmi,j→i ≡
   𝔼{}→{i}[0, 0, 1]]
Out[∘]= {3.6875, True}
```

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= Timing@Block[{$k = 2},
$\left(\left(\left(R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}\right) \sim B_i \sim dS_i\right) \left(R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j}\right)\right) \sim B_{i,j} \sim dm_{i,j\to i}$]

Out[∘]= $\left\{4.85938, \mathbb{E}_{\{\}\to\{i\}}\left[0, 0, \frac{1}{B_i} + \frac{\hbar\, a_i\, \epsilon}{B_i} + \frac{\hbar^2\, a_i^2\, \epsilon^2}{2\, B_i} + O[\epsilon]^3\right]\right\}$

In[∘]:= Timing@Block[{$k = 2}, HL /@ {$\left(C_i \bar{C}_j\right) \sim B_{i,j} \sim dm_{i,j\to i} \equiv \mathbb{E}_{\{\}\to\{i\}}[0, 0, 1]$, $\left(\bar{C}_i \bar{C}_j\right) \sim B_{i,j} \sim dm_{i,j\to i} \equiv$
$\left(\left(\left(R_{1,2} \sim B_1 \sim dS_1 \sim B_{1,2} \sim dm_{2,1\to i}\right) \sim B_i \sim dS_i\right) \left(R_{1,2} \sim B_2 \sim dS_2 \sim B_2 \sim dS_2 \sim B_{1,2} \sim dm_{2,1\to j}\right)\right) \sim B_{i,j} \sim dm_{i,j\to i}$}]

Out[∘]= {5.59375, {True, True}}

Reidemeister 2:

In[∘]:= Timing[HL[# ≡ $\mathbb{E}_{\{\}\to\{1,2\}}$[0, 0, 1]] & /@
{$\left(\bar{R}_{1,2} R_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)$, $\left(R_{1,2} \bar{R}_{3,4}\right) \sim B_{1,2,3,4} \sim \left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)$}]

Out[∘]= {4.78125, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= Timing@HL[$\left(R_{1,4} \bar{R}_{5,2} \bar{C}_3\right) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \bar{C}_1\, \mathbb{E}_{\{\}\to\{2\}}[0, 0, 1]$]

Out[∘]= {2.73438, True}

Reidemeister 3:

In[∘]:= Timing@HL[$\left(\left(R_{1,2} R_{4,3} R_{5,6}\right) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\right) \equiv$
$\left(\left(R_{1,6} R_{2,3} R_{4,5}\right) \sim B_{1,4} \sim dm_{1,4\to 1} \sim B_{2,5} \sim dm_{2,5\to 2} \sim B_{3,6} \sim dm_{3,6\to 3}\right)$]

Out[∘]= {5.9375, True}

Relations between the four kinks:

In[∘]:= Timing[HL /@ {$Kink_i \equiv \left(R_{3,1} C_2\right) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i}$,
$\overline{Kink}_j \equiv \left(\bar{R}_{3,1} \bar{C}_2\right) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j}$, $\left(Kink_i\, \overline{Kink}_j\right) \sim B_{i,j} \sim dm_{i,j\to 1} \equiv \mathbb{E}_{\{\}\to\{1\}}[0, 0, 1]$}]

Out[∘]= {5.23438, {True, True, True}}

The Trefoil

In[∘]:= Timing@Block[{$k = 1},
$Z = R_{1,5} R_{6,2} R_{3,7} \bar{C}_4 \overline{Kink}_8 \overline{Kink}_9 \overline{Kink}_{10}$;
Do[$Z = Z \sim B_{1,r} \sim dm_{1,r\to 1}$, {r, 2, 10}];
{Simplify /@ Z, Simplify /@ $\left(Z \sim B_1 \sim b2t_1 \text{ /. } T_1 \to T\right)$}]

Out[∘]= $\left\{10.0313, \left\{\mathbb{E}_{\{\}\to\{1\}}\left[0, 0, \frac{B_1}{1 - B_1 + B_1^2} - \left(\hbar\, B_1 \left(-a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) + \gamma \left(B_1 - 2 B_1^2 - 2 B_1^4 + 2 \hbar\, x_1 y_1 + B_1^3 \left(3 + 2 \hbar\, x_1 y_1\right)\right)\right) \epsilon\right) \middle/ \left(1 - B_1 + B_1^2\right)^3 + O[\epsilon]^2\right], \mathbb{E}_{\{\}\to\{1\}}\left[0, 0, \frac{T}{1 - T + T^2} + \left(T \hbar \left(T \left(-1 + 2 T - 3 T^2 + 2 T^3\right) \gamma + 2 \left(-1 + T - T^3 + T^4\right) a_1 - 2 \left(1 + T^3\right) \gamma \hbar\, x_1 y_1\right) \epsilon\right) \middle/ \left(1 - T + T^2\right)^3 + O[\epsilon]^2\right]\right\}\right\}$

Program

```
In[∘]:= Define[kR_{i,j} = R_{i,j} // (b2t_i b2t_j) /. t_{i|j} → t,
  k̅R̅_{i,j} = R̅_{i,j} // (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
  km_{i,j→k} = (t2b_i t2b_j) // dm_{i,j→k} // b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kC_i = C_i // b2t_i /. T_i → T,
  k̅C̅_i = C̅_i // b2t_i /. T_i → T,
  kKink_i = Kink_i // b2t_i /. {t_i → t, T_i → T},
  k̅K̅i̅n̅k̅_i = K̅i̅n̅k̅_i // b2t_i /. {t_i → t, T_i → T}]
```

```
In[∘]:= Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} k̅C̅_4 k̅K̅i̅n̅k̅_8 k̅K̅i̅n̅k̅_9 k̅K̅i̅n̅k̅_10;
  Do[Z = Z~B_{1,r}~km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

Out[∘]=

$$\left\{5.5625,\ \mathbb{E}_{\{\}\to\{1\}}\left[0,\ 0,\ \frac{T}{1-T+T^2}+\left(T\hbar\left(T\left(-1+2T-3T^2+2T^3\right)\gamma+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)\gamma\hbar x_1 y_1\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+O[\epsilon]^2\right]\right\}$$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

Program

```
In[∘]:= RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

Program

```
In[∘]:= RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front = {0}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = Cases[pd, x_X :→ { Xp[x⟦4⟧, x⟦1⟧]   PositiveQ@x
                           { Xm[x⟦2⟧, x⟦1⟧]   True       ];
    For[k = 0, k < 2 n, ++k, If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → (xs /. {
            Xp[k + 1, l_] | Xm[l_, k + 1] :→ {l, k + 1, 1 - l},
            Xp[l_, k + 1] | Xm[k + 1, l_] :→ (++rots⟦l⟧; {1 - l, k + 1, l})
           })],
      Cases[front, k | -k] /. {k, -k} :→ --rots⟦k + 1⟧;
     ]];
    RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

```
In[∘]:= xs = Cases[pd, x_X :→ If[PositiveQ@x, Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]]];
```

In[ ]:=
```
RVK[Knot[10, 100]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[ ]=
```
RVK[{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8],
  Xm[19, 12], Xp[17, 2]}, {0, 0, 0, 0, 0, -1, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0}]
```

Program

In[ ]:=
```
rot[i_, 0] := E_{{}→{i}}[0, 0, 1];
rot[i_, n_] := Module[{j},
   rot[i, n] = If[n > 0, rot[i, n - 1] kC_j, rot[i, n + 1] kC̄_j] // km_{i,j→i}];
```

Program

In[ ]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
 PP"Z"@Module[{todo, n, rots, ζ, done, st, cx, ζ1, i, j, k, k1, k2, k3},
   {todo, rots} = List @@ rvk;
   AppendTo[rots, 0];
   n = Length[todo];
   ζ = E_{{}→{0}}[0, 0, 1];
   done = {0};
   st = Range[0, 2 n + 1];
   While[Echo@todo =!= {},
    {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
    {i, j} = List @@ cx;
    ζ1 = Switch[Head[cx],
       Xp, (kR_{i,j} kKink̄_k) // km_{j,k→j},
       Xm, (kR̄_{i,j} kKink_k) // km_{j,k→j}
      ];
    ζ1 = (rot[k, rots[[i]]] ζ1) // km_{k,i→i}; rots[[i]] = 0;
    ζ1 = (ζ1 rot[k, rots[[i + 1]]]) // km_{i,k→i}; rots[[i + 1]] = 0;
    ζ1 = (rot[k, rots[[j]]] ζ1) // km_{k,j→j}; rots[[j]] = 0;
    ζ1 = (ζ1 rot[k, rots[[j + 1]]]) // km_{j,k→j}; rots[[j + 1]] = 0;
    ζ *= ζ1;
    If[MemberQ[done, i], ζ = ζ // km_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
    If[MemberQ[done, i - 1], ζ = ζ // km_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
    If[MemberQ[done, j], ζ = ζ // km_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
    If[MemberQ[done, j - 1], ζ = ζ // km_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
    done = done ⋃ {i - 1, i, j - 1, j};
    todo = DeleteCases[todo, cx]
   ];
   Simplify /@ (ζ /. {x_0 → x, y_0 → y, a_0 → a})
  ]
```

Knot

In[ ]:=
```
$k = 1; Timing@Z@Knot[10, 100]
```

Knot

```
{Xp[1, 6], Xp[5, 18], Xm[13, 20], Xm[7, 14],
 Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xp[5, 18], Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12], Xp[17, 2]}
```

Knot

```
{Xm[13, 20], Xm[7, 14], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[15, 8], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[9, 16], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[3, 10], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[11, 4], Xm[19, 12]}
```

Knot

```
{Xm[13, 20], Xm[19, 12]}
```

Knot

```
{Xm[13, 20]}
```

Knot

```
{}
```

Knot

Out[∘]=

$$\Big\{181.516,\ \mathbb{E}_{\{\}\to\{0\}}\Big[0,\ 0,\ T^4 \big/ \left(1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8\right) + \Big(T^4\hbar\,\Big(4a\left(-2+14T-51T^2+120T^3-203T^4+258T^5-246T^6+152T^7-152T^9+246T^{10}-258T^{11}+203T^{12}-120T^{13}+51T^{14}-14T^{15}+2T^{16}\right) + \gamma\Big(-6+2T^{16}-8xy\hbar-440T^9\left(-1+xy\hbar\right)-4T^{15}\left(3+2xy\hbar\right)+8T^8\left(-97+21xy\hbar\right)+8T^7\left(131+21xy\hbar\right)-20T^6\left(57+22xy\hbar\right)+T^{14}\left(37+48xy\hbar\right)+T\left(44+48xy\hbar\right)-8T^{11}\left(2+61xy\hbar\right)+8T^5\left(127+68xy\hbar\right)-2T^{13}\left(35+78xy\hbar\right)+4T^{10}\left(-39+136xy\hbar\right)-T^2\left(167+156xy\hbar\right)+T^{12}\left(79+324xy\hbar\right)+T^3\left(410+324xy\hbar\right)-T^4\left(733+488xy\hbar\right)\Big)\Big)\,\epsilon\Big) \big/ \left(1-4T+9T^2-12T^3+13T^4-12T^5+9T^6-4T^7+T^8\right)^3 + O[\epsilon]^2\Big]\Big\}$$

In[∘]:= **EndProfile**[];

Profile

In[∘]:= **PrintProfile**[]

Profile

Out[∘]=

```
ProfileRoot is root. Profiled time: 311.48
  (     1)    0.172/ 181.500 above Z
  (   157)    0.561/ 111.572 above B
  (    37)    0.141/  18.253 above Boot
  (   147)    0.046/   0.124 above CF
  (     2)    0.031/   0.031 above RVK
CF: called 13049 times, time in 76.079/208.54
  (  2694)   31.909/ 144.620 under LZip
  (   147)    0.046/   0.124 under ProfileRoot
  ( 10208)   44.124/  63.796 under QZip
  ( 39202)   35.797/ 132.461 above CCF
LZip: called 294 times, time in 64.246/218.808
  (   294)   64.246/ 218.808 under B
```

```
  (  2694)   31.909/ 144.620 above CF
  (   294)    1.879/   9.942 above Zip
Together: called 160474 times, time in 56.495/112.293
  ( 160474)   56.495/ 112.293 under CCF
  ( 160474)   20.258/  55.798 above Exp
CCF: called 160474 times, time in 55.708/168.001
  ( 39202)   35.797/ 132.461 under CF
  ( 121272)   19.911/  35.540 under Exp
  ( 160474)   56.495/ 112.293 above Together
QZip: called 294 times, time in 20.514/91.236
  (   294)   20.514/  91.236 under B
  ( 10208)   44.124/  63.796 above CF
  (   294)    1.565/   6.926 above Zip
Exp: called 160474 times, time in 20.258/55.798
  ( 160474)   20.258/  55.798 under Together
  ( 121272)   19.911/  35.540 above CCF
Zip: called 2675 times, time in 13.889/55.743
  (   294)    1.879/   9.942 under LZip
  (   294)    1.565/   6.926 under QZip
  (  2087)   10.445/  38.875 under Zip
  (  2675)    2.979/   2.979 above Collect
  (  2087)   10.445/  38.875 above Zip
Collect: called 2675 times, time in 2.979/2.979
  (  2675)    2.979/   2.979 under Zip
B: called 294 times, time in 0.749/310.793
  (    72)    0.141/ 181.140 under Z
  (    65)    0.047/  18.081 under Boot
  (   157)    0.561/ 111.572 under ProfileRoot
  (   294)   64.246/ 218.808 above LZip
  (   294)   20.514/  91.236 above QZip
Boot: called 59 times, time in 0.36/25.582
  (     3)       0/   0.188 under Z
  (    19)    0.219/   7.141 under Boot
  (    37)    0.141/  18.253 under ProfileRoot
  (    65)    0.047/  18.081 above B
  (    19)    0.219/   7.141 above Boot
Z: called 1 times, time in 0.172/181.5
  (     1)    0.172/ 181.500 under ProfileRoot
  (    72)    0.141/ 181.140 above B
  (     3)       0/   0.188 above Boot
RVK: called 2 times, time in 0.031/0.031
  (     2)    0.031/   0.031 under ProfileRoot
```